Simulating Predator-Prey Dynamics with Visualization

Tanuj Karthikeyan
January 16th 2025

This Mathematica script is meant to simulate predator prey relationships and population amounts with adjustable sliders for the different variables of the Lotka-Volterra equations and evaluating it over time.

2D Overlapping Graph

In[∘]:=
```
Manipulate[Module[{solution, preyPopulation, predatorPopulation},
  solution = NDSolve[{x'[t] == α x[t] - β x[t] × y[t],
     y'[t] == δ x[t] × y[t] - γ y[t], x[0] == x0, y[0] == y0}, {x, y}, {t, 0, tmax}];
  preyPopulation = x[t] /. solution[[1]];
  predatorPopulation = y[t] /. solution[[1]];
  Show[Plot[{preyPopulation, predatorPopulation}, {t, 0, tmax},
    PlotLegends → {"Prey Population Size", "Predator Population Size"},
    PlotStyle → {{Blue, Thick}, {Red, Thick}}, AxesLabel → {"Time", "Population"},
    PlotRange → All, PlotLabel → Style["2D Predator-Prey Dynamics", Bold, 14]]]],
 {{α, 0.5, "Prey Growth Rate"}, 0.1, 1, 0.01},
 {{β, 0.02, "Predation Rate"}, 0.01, 0.1, 0.001},
 {{γ, 0.5, "Predator Death Rate"}, 0.1, 1, 0.01},
 {{δ, 0.02, "Predator Reproduction Rate"}, 0.01, 0.1, 0.001},
 {{x0, 40, "Initial Prey Population"}, 10, 100, 1},
 {{y0, 9, "Initial Predator Population"}, 1, 50, 1},
 {{tmax, 50, "Simulation Time"}, 10, 100, 1}]
```

Out[∘]=



This graph is fully customizable as the user is able to adjust the slider to determine the variables in the Lotka-Volterra equations resulting in completely different graphs.

3D Graph

In[ ]:=
```
Manipulate[Module[{solution, preyPop, predatorPop},
  solution = NDSolve[{x'[t] == α x[t] - β x[t] × y[t],
     y'[t] == δ x[t] × y[t] - γ y[t], x[0] == x0, y[0] == y0}, {x, y}, {t, 0, tmax}];
  preyPop = x[t] /. solution[[1]];
  predatorPop = y[t] /. solution[[1]];
  (*Generate 3D Parametric Plot*)
  ParametricPlot3D[{preyPop, predatorPop, t}, {t, 0, tmax}, PlotStyle → Thick,
   AxesLabel → {"Prey Population", "Predator Population", "Time"}, PlotRange → All,
   Boxed → True, TicksStyle → Directive[Black, Bold], ViewPoint → {1.3, -2.4, 1.8},
   PlotLabel → Style["3D Predator-Prey Dynamics", Bold, 14],
   MeshFunctions → {#3 &}, Mesh → 10, MeshStyle → Directive[Gray, Dashed]]],
 (*Sliders for Parameters*){{α, 0.5, "Prey Growth Rate"}, 0.1, 1, 0.01},
 {{β, 0.02, "Predation Rate"}, 0.005, 0.05, 0.001},
 {{γ, 0.5, "Predator Death Rate"}, 0.1, 1, 0.01},
 {{δ, 0.02, "Predator Reproduction Rate"}, 0.005, 0.05, 0.001},
 {{x0, 40, "Initial Prey Population"}, 10, 100, 1},
 {{y0, 9, "Initial Predator Population"}, 1, 50, 1},
 {{tmax, 50, "Time Duration"}, 10, 100, 5}]
```

Out[ ]=